

elisp literate library

a literate programming tool to write Emacs lisp codes in org mode.

Jingtao Xu

June 9, 2019

Contents

1	Introduction	1
2	How to do it?	1
3	Implementation	2
3.1	Preparation	2
3.2	stream read functions	2
3.3	handle org mode syntax	4
3.4	load/compile org file with new syntax	8
3.5	function to tangle org file to elisp file	10
4	Tests	12
4.1	Introduction	12
4.2	test cases	12
5	References	13

1 Introduction

An Emacs library or configuration file can write in org mode then tangle to an elisp file later, here is one example: [Emacs configurations written in Org mode](#) .

But What if I want to write a library or a configuration file in org file and load it to Emacs directly? If it can, then we will have an uniform development environment without keeping multiple copies of codes. Furthermore, we can jump to the elisp definition in an org file directly when required. That will be a convenient way for our daily development.

So is this library, which extends the Emacs [load](#) mechanism so Emacs can load org files as lisp source files directly.

2 How to do it?

In org mode, the Emacs lisp codes surround by lines between `#+begin_src elisp` and `#+end_src` (see [org manual](#)).

```
#+BEGIN_SRC elisp :load no
(message "this is a test.~%")
#+END_SRC
```

So to let Emacs lisp can read an org file directly, all lines out of surrounding by `#+begin_src elisp` and `#+end_src` should mean nothing, and even codes surrounding by them should mean nothing if the [header arguments](#) in a code block request such behavior.

Here is a trick, a new Emacs lisp reader function get implemented (by binding elisp variable [load-read-function](#)) to replace original `read` function when using elisp function `load` to load a org file.

The new reader will make elisp reader enter into org mode syntax, which means it will ignore all lines until it meet `#+BEGIN_SRC elisp`.

When `#+begin_src elisp` occur, [header arguments](#) for this code block will give us a chance to switch back to normal Emacs lisp reader or not.

And if it switch back to normal Emacs lisp reader, the end line `#+END_SRC` should mean the end of current code block, if it occur, then the reader will switch back to org mode syntax. if not, then the reader will continue to read subsequent stream as like the original Emacs lisp reader.

3 Implementation

3.1 Preparation

We use common lisp macros in this library

```
(require 'cl-lib)
```

And advice facility library.

```
(require 'nadvice)
```

There is a debug variable to switch on/off the log messages for this library.

```
(defvar literate-elisp-debug-p nil)
```

There is also a dynamic Boolean variable bounded by our read function while parsing is in progress. It'll indicate whether org mode syntax or elisp mode syntax is in use.

```
(defvar literate-elisp-org-code-blocks-p nil)
```

And the code block begin/end identifies:

```
(defvar literate-elisp-begin-src-id "+BEGIN_SRC elisp")
(defvar literate-elisp-end-src-id "+END_SRC")
```

3.2 stream read functions

To give us the ability of syntax analysis, stream read actions such as peek a character or read and drop next character should get implemented.

The `input streams` are the same streams used by the original elisp `read` function.

3.2.1 literate-elisp-peek

```
(defun literate-elisp-peek (in)
  "Return the next character without dropping it from the stream.
Argument IN: input stream."
  (cond ((bufferp in)
        (with-current-buffer in
          (when (not (eobp))
            (char-after))))
        ((markerp in)
        (with-current-buffer (marker-buffer in)
          (when (< (marker-position in) (point-max))
            (char-after in))))
        ((functionp in)
        (let ((c (funcall in)))
          (when c
            (funcall in c)
            c))))
```

3.2.2 literate-elisp-next

```
(defun literate-elisp-next (in)
  "Given a stream function, return and discard the next character.
Argument IN: input stream."
  (cond ((bufferp in)
        (with-current-buffer in
          (when (not (eobp))
            (progn
              (char-after)
              (char-after))))
        ((functionp in)
        (let ((c (funcall in)))
          (when c
            (funcall in c)
            c))))
```

```

        (forward-char 1))))))
    (markerp in)
    (with-current-buffer (marker-buffer in)
      (when (< (marker-position in) (point-max))
        (progl
          (char-after in)
          (forward-char 1))))))
    ((functionp in)
     (funcall in))))

```

3.2.3 literate-elisp-position

This functions is a helpful function to debug our library.

```

(defun literate-elisp-position (in)
  "Return the current position from the stream.
Argument IN: input stream."
  (cond ((bufferp in)
         (with-current-buffer in
           (point)))
        ((markerp in)
         (with-current-buffer (marker-buffer in)
           (marker-position in)))
        ((functionp in)
         "Unknown")))

```

3.2.4 literate-elisp-read-until-end-of-line

when read org file character by character, if current line determines as an org syntax, then the whole line should ignore, so there should exist such a function.

Before then, let's implement an abstract method to read characters repeatedly while a predication meet.

The ignored string return from this function because it may be useful sometimes, for example when reading `header arguments` after `#+begin_src elisp`.

```

(defun literate-elisp-read-while (in pred)
  "Read and return a string from the input stream, as long as the predicate.
Argument IN: input stream.
Argument PRED: predicate function."
  (let ((chars (list)) ch)
    (while (and (setq ch (literate-elisp-peek in))
                (funcall pred ch))
      (push (literate-elisp-next in) chars))
    (apply #'string (nreverse chars))))

```

Now reading until end of line is easy to implement.

```

(defun literate-elisp-read-until-end-of-line (in)
  "Skip over a line (move to 'end-of-line')."
  Argument IN: input stream."
  (progl
    (literate-elisp-read-while in (lambda (ch)

```

```
(not (eq ch ?\n)))
(literate-elisp-next in)))
```

3.3 handle org mode syntax

3.3.1 code block header argument load

There are a lot of different elisp codes occur in one org file, some for function implementation, some for demo, and some for test, so an **org code block header argument load** to decide to read them or not should define, and it has two meanings:

- yes
It means that current code block should load normally, it is the default mode when the header argument load is not provided.
- no
It means that current code block should ignore by elisp reader.
- test
It means that current code block should load only when variable `literate-elisp-test-p` is true.

```
(defvar literate-elisp-test-p nil)
```

Now let's implement above rule.

```
(defun literate-elisp-load-p (flag)
  "Load current elisp code block or not.
Argument FLAG: flag symbol."
  (cl-case flag
    ((yes nil) t)
    (test literate-elisp-test-p)
    (no nil)
    (t nil)))
```

Let's also implement a function to read **header arguments** after `#+BEGIN_SRC elisp`, and convert every key and value to a elisp symbol (test is here: [4.2.3](#)).

```
(defun literate-elisp-read-header-arguments (arguments)
  "Read org code block header arguments.
Argument ARGUMENTS: a string to hold the arguments."
  (cl-loop for token in (split-string arguments)
    collect (intern token)))
```

Let's define a convenient function to get load flag from the input stream.

```
(defun literate-elisp-get-load-option (in)
  "Read load option from input stream.
Argument IN: input stream."
```

```
(cl-getf (literate-elisp-read-header-arguments (literate-elisp-read-until-end-of-line
↪ in)) :load))
```

3.3.2 fix of invalid-read-syntax

Emacs original `read` function will try to skip all comments until it can get a valid elisp form, so when we call original `read` function and there are no valid elisp form left in one code block, it may reach `#+end_src`, as it don't know how to read it, it will signal an error description (`invalid-read-syntax "#`"). So when such error occur, we have to handle it(test is here:[4.2.1](#)).

Please note that the stream position is just after the character `#` when above error occur.

```
(defmacro literate-elisp-fix-invalid-read-syntax (in &rest body)
  "Fix read error 'invalid-read-syntax'.
Argument IN: input stream.
Argument BODY: body codes."
  `(condition-case ex
    ,@body
    (invalid-read-syntax
     (when literate-elisp-debug-p
      (message "reach invalid read syntax %s at position %s"
               ex (literate-elisp-position in)))
     (if (equal "#" (second ex))
         ;; maybe this is #+end_src
         (literate-elisp-read-after-sharpsign in)
         ;; re-throw this signal because we don't know how to handle it.
         (signal (car ex) (cdr err))))))
```

3.3.3 handle prefix spaces.

Sometimes `#+begin_src` elisp and `#+end_src` may have prefix spaces, let's ignore them carefully.

If it is not processed correctly, the reader may enter into an infinite loop, especially when using a custom reader to tangle codes.

```
(defun literate-elisp-ignore-white-space (in)
  "Skip white space characters.
Argument IN: input stream."
  (while (cl-find (literate-elisp-peek in) '(?\n ?\ ?\t))
    ;; discard current character.
    (literate-elisp-next in))
```

3.3.4 alternative elisp read function

When tangling org file, we want to tangle elisp codes without changing them(but Emacs original `read` will), so let's define a variable

to hold the actual elisp reader used by us then it can be changed when tangling org files(see 3.5).

```
(defvar literate-elisp-read (symbol-function 'read))
```

We don't use the original symbol `read` in `literate-elisp-read` because sometimes function `read` can be changed by the following elisp code

```
(fset 'read (symbol-function 'literate-elisp-read-internal))
```

So we can ensure that `literate-elisp-read` will always use the original `read` function, which will not be altered when we want to byte compile the org file by function `literate-elisp-byte-compile-file`.

3.3.5 basic read routine for org mode syntax.

It's time to implement the main routine to read literate org file. The basic idea is simple, ignoring all lines out of elisp source block, and be careful about the special character `#`.

```
(defun literate-elisp-read-datum (in)
  "Read and return a Lisp datum from the input stream.
Argument IN: input stream."

  (literate-elisp-ignore-white-space in)
  (let ((ch (literate-elisp-peek in)))
    (when literate-elisp-debug-p
      (message "literate-elisp-read-datum to character '%c' (position:%s)."
               ch (literate-elisp-position in))))

  (literate-elisp-fix-invalid-read-syntax in
    (cond
      ((not ch)
       (error "End of file during parsing"))
      ((and (not literate-elisp-org-code-blocks-p)
            (not (eq ch ?\#)))
       (let ((line (literate-elisp-read-until-end-of-line in)))
         (when literate-elisp-debug-p
           (message "ignore line %s" line)))
        nil)
      ((eq ch ?\#)
       (literate-elisp-next in)
       (literate-elisp-read-after-sharpsign in))
      (t (funcall literate-elisp-read in))))))
```

3.3.6 how to handle when meet

We have to be carefully when meeting the character `#` and handle different conditions that may occur:

```
(defun literate-elisp-read-after-sharpsign (in)
  "Read after #.
Argument IN: input stream."
```

```

;; if it is not inside an elisp syntax
(cond ((not literate-elisp-org-code-blocks-p)
  ;; check if it is `#+begin_src elisp'
  (if (cl-loop for i from 1 below (length literate-elisp-begin-src-id)
    for c1 = (aref literate-elisp-begin-src-id i)
    for c2 = (literate-elisp-next in)
    thereis (not (char-equal c1 c2)))
    ;; if it is not, continue to use org syntax and ignore this line
    (progn (literate-elisp-read-until-end-of-line in)
      nil)
    ;; if it is, read source block header arguments for this code block and
    ↪ check if it should be loaded.
    (cond ((literate-elisp-load-p (literate-elisp-get-load-option in))
      ;; if it should be loaded, switch to elisp syntax context
      (when literate-elisp-debug-p
        (message "enter into a elisp code block"))
        (setf literate-elisp-org-code-blocks-p t)
        nil)
      (t
        ;; if it should not be loaded, continue to use org syntax and ignore
        ↪ this line
        nil))))))

(t
  ;; 2. if it is inside an elisp syntax
  (let ((c (literate-elisp-next in)))
    (when literate-elisp-debug-p
      (message "found %#c inside a org block" c))
    (cl-case c
      ;; check if it is ~#+~, which has only legal meaning when it is equal `#+
      ↪ end_src'
      (?\+
        (let ((line (literate-elisp-read-until-end-of-line in)))
          (when literate-elisp-debug-p
            (message "found org elisp end block:%s" line)))
        ;; if it is, then switch to org mode syntax.
        (setf literate-elisp-org-code-blocks-p nil)
        nil)
      ;; if it is not, then use original elisp reader to read the following
      ↪ stream
      (t (funcall literate-elisp-read in))))))

```

3.4 load/compile org file with new syntax

3.4.1 literate reader is in use when loading a org file

original function read will read until it can get a valid lisp form, we will try to keep this behavior.

```

(defun literate-elisp-read-internal (&optional in)
  "A wrapper to follow the behavior of original read function.
Argument IN: input stream."
  (cl-loop for form = (literate-elisp-read-datum in)
    if form
      do (cl-return form)
      ;; if original read function return nil, just return it.
    if literate-elisp-org-code-blocks-p
      do (cl-return nil)
      ;; if it reach end of stream.
    if (null (literate-elisp-peek in))

```



```
do (cl-return nil)))
```

Now we define the `literate read` function which will bind to Emacs variable `load-read-function`.

```
(defun literate-elisp-read (&optional in)
  "Literate read function.
Argument IN: input stream."
  (if (and load-file-name
        (string-match "\\..org\\'" load-file-name))
      (literate-elisp-read-internal in)
      (read in)))
```

And the main exported function to do `literate load`.

```
(defun literate-elisp-load (path)
  "Literate load function.
Argument PATH: target file to load."
  (let ((load-read-function (symbol-function 'literate-elisp-read))
        (literate-elisp-org-code-blocks-p nil))
    (load path)))
```

If you want to `literate load` file in batch mode, here it is:

```
(defun literate-elisp-batch-load ()
  "Literate load file in 'command-line' arguments."
  (or noninteractive
      (signal 'user-error' ("This function is only for use in batch mode"))))
  (if command-line-args-left
      (literate-elisp-load (pop command-line-args-left))
      (error "No argument left for 'literate-elisp-batch-load'")))
```

3.4.2 an interactive command to load a `literate org` file from Emacs

```
(defun literate-elisp-load-file (file)
  "Load the Lisp file named FILE.
Argument FILE: target file path."
  ;; This is a case where .elc and .so/.dll make a lot of sense.
  (interactive (list (read-file-name "Load org file: " nil nil 'lambda)))
  (literate-elisp-load (expand-file-name file)))
```

3.4.3 a function to byte compile a `literate org` file

Currently(2018.12.16) Emacs `bytecomp` library always use function `read` to read elisp forms, instead of the function specified by variable `load-read-function`.so we modify the symbol function of `read` when byte compiling `org` file.

```
(defun literate-elisp-byte-compile-file (file &optional load)
  "Byte compile an org file.
Argument FILE: file to compile.
Argument LOAD: load the file after compiling."
  (interactive
   (let ((file buffer-file-name)
```

```

        (file-dir nil))
    (and file
      (derived-mode-p 'org-mode)
      (setq file-dir (file-name-directory file)))
    (list (read-file-name (if current-prefix-arg
                              "Byte compile and load file: "
                              "Byte compile file: ")
                          file-dir buffer-file-name nil)
          current-prefix-arg)))
  (let ((iterate-elisp-org-code-blocks-p nil)
        (load-file-name buffer-file-name)
        (original-read (symbol-function 'read)))
    (fset 'read (symbol-function 'iterate-elisp-read-internal))
    (unwind-protect
      (byte-compile-file file load)
      (fset 'read original-read))))

```

After byte compiling an literate org file, it will be compiled to a file with suffix `.org.elc`, after loading such compiled file, Emacs will fail to find the variable or function definition because function `find-library-name` don't treat org file as a source file, so we have to add an advice function to `find-library-name` to fix this issue.

```

(defun iterate-elisp-find-library-name (orig-fun &rest args)
  "An advice to make 'find-library-name' can recognize org source file.
Argument ORIG-FUN: original function of this advice.
Argument ARGS: the arguments to original advice function."

  (when (string-match "\\(\\.org\\.el\\)" (car args))
    (setf (car args) (replace-match ".org" t t (car args)))
    (when iterate-elisp-debug-p
      (message "fix literate compiled file in find-library-name :%s" (car args))))
  (apply orig-fun args))
(advice-add 'find-library-name :around #'iterate-elisp-find-library-name)

```

3.5 function to tangle org file to elisp file

To build an Emacs lisp file from an org file without depending on `iterate-elisp` library, we need tangle an org file to an Emacs lisp file(.el).

Firstly, when tangle elisp codes, we don't want to use original Emacs read function to read them because it will ignore comment lines and it's hard for us to revert them back to a pretty print code, so we define a new reader function and bind it to variable `iterate-elisp-read`.

This reader will read codes in a code block without changing them until it reach `#+end_src`.

```

(defun iterate-elisp-tangle-reader (&optional buf)
  "Tangling codes in one code block.
Argument BUF: source buffer."
  (with-output-to-string
    (with-current-buffer buf
      (when (/= (point) (line-beginning-position))
        ;; if reader still in last line,move it to next line.

```

```

(forward-line 1))

(loop for line = (buffer-substring-no-properties (line-beginning-position) (
  ↳ line-end-position))
      until (or (eobp)
                (string-equal (trim-string (downcase line)) "#+end_src")))
  do (loop for c across line
          do (write-char c))
      (when literate-elisp-debug-p
        (message "tangle elisp line %s" line))
      (write-char ?\n)
      (forward-line 1))))

```

Now we can tangle the elisp code blocks with the following codes.

```

(cl-defun literate-elisp-tangle (file &key (el-file (concat (file-name-sans-extension
  ↳ file) ".el")))
  (let* ((source-buffer (find-file-noselect file))
         (target-buffer (find-file-noselect el-file))
         (org-path-name (concat (file-name-base file) "." (file-name-extension file)))
         (literate-elisp-read 'literate-elisp-tangle-reader)
         (literate-elisp-test-p test-p)
         (literate-elisp-org-code-blocks-p nil))
    (with-current-buffer target-buffer
      (delete-region (point-min) (point-max))
      (when header
        (insert header "\n"))
      (insert ";;; Code:\n\n"
              ";;; The code is automatically generated by function `
              ↳ literate-elisp-tangle' from file '" org-path-name "'.\n"
              ";;; It is not designed to be readable by a human.\n"
              ";;; It is generated to load by Emacs directly without depending on `
              ↳ literate-elisp'.\n"
              ";;; you should read file '" org-path-name "' to find out the usage and
              ↳ implementation detail of this source file.\n\n"
              "\n"))

      (with-current-buffer source-buffer
        (goto-char (point-min))
        (cl-loop for obj = (literate-elisp-read-internal source-buffer)
                  if obj
                    do (with-current-buffer target-buffer
                        (insert obj "\n"))
                    until (eobp)))

      (with-current-buffer target-buffer
        (when tail
          (insert "\n" tail))
        (save-buffer)
        (kill-current-buffer))))

```

And when a new version of `./literate-elisp.el` can release from this file, the following code should execute.

```

(literate-elisp-tangle
 "literate-elisp.org"

```

```

:header ";;; literate-elisp.el --- literate program to write elisp codes in org mode
↳ -- lexical-binding: t; --

;; Copyright (C) 2018-2019 Jingtao Xu

;; Author: Jingtao Xu <jingtaozf@gmail.com>
;; Created: 6 Dec 2018
;; Version: 0.1
;; Keywords: lisp docs extensions tools
;; URL: https://github.com/jingtaozf/literate-elisp
;; Package-Requires: ((cl-lib "0.6") (emacs "24.4"))

;; This program is free software; you can redistribute it and/or modify
;; it under the terms of the GNU General Public License as published by
;; the Free Software Foundation, either version 3 of the License, or
;; (at your option) any later version.

;; This program is distributed in the hope that it will be useful,
;; but WITHOUT ANY WARRANTY; without even the implied warranty of
;; MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the
;; GNU General Public License for more details.

;; You should have received a copy of the GNU General Public License
;; along with this program. If not, see <http://www.gnu.org/licenses/>.

;;; Commentary:

;; Literate-elisp is an Emacs Lisp library to provide an easy way to use literate
;; ↳ programming in Emacs Lisp.
;; It extends the Emacs load mechanism so Emacs can load Org files as Lisp source files
;; ↳ directly.
"
          :tail "(provide 'literate-elisp)
;;; literate-elisp.el ends here
")

```

The head and tail lines require by [MELPA](#) repository.
Now let's check the elisp file to meet the requirement of [MELPA](#).

```

(with-current-buffer (find-file "literate-elisp.el")
  (checkdoc)
  (package-lint-current-buffer))

```

4 Tests

4.1 Introduction

We use [ERT](#) library to define and run tests. Web service [travis ci](#) will load config file [./travis.yml](#) to run these tests automatically every time there is a new git change.

4.2 test cases

4.2.1 test the empty code block

If one code block is empty, we will use Emacs original `read` function, which will read `#+end_src` and signal an error, let's test whether `literate-elisp` can read it gracefully.

```
;; This is a comment line to test empty code block.
```

4.2.2 test code block with prefix space.

Some code block have white spaces before `#+begin_src` elisp, let's test whether `literate-elisp` can read it normally.

```
(defvar literate-elisp-a-test-variable 10)
```

Let's write a test case for above code block.

```
(ert-deftest literate-elisp-read-code-block-with-prefix-space ()  
  "A spec of code block with prefix space."  
  (should (equal literate-elisp-a-test-variable 10)))
```

4.2.3 test `literate-elisp-read-header-arguments`

```
(ert-deftest literate-elisp-read-header-arguments ()  
  "A spec of function to read org header-arguments."  
  (should (equal (literate-elisp-read-header-arguments " :load yes") '(:load yes)))  
  (should (equal (literate-elisp-read-header-arguments " :load no ") '(:load no)))  
  (should (equal (literate-elisp-read-header-arguments ":load yes") '(:load yes))))
```

5 References

- [Literate. Programming.](#) by Donald E. Knuth
- [Literate Programming](#) a site of literate programming
- [Literate Programming in the Large](#) a talk video from Timothy Daly, one of the original authors of [Axiom](#).
- [literate programming in org babel](#)
- [A collection of literate programming examples using Emacs Org mode](#)
- [elisp-reader.el](#) customized reader for Emacs Lisp